

Paper 340-2011

Modeling Financial Risk Factor Correlation with the COPULA Procedure

Jan Chvosta, Donald J. Erdman, Mark Little
SAS Institute Inc., Cary, NC

ABSTRACT

Two keys to measuring and controlling the risk inherent in financial securities are (1) understanding the volatility of economic factors on which the value of the portfolio depends and (2) understanding how changes in those economic factors are related to each other. Recent progress in the mathematical technique of “copula” functions offers a powerful new approach to modeling dependencies among numerous risk factors. This paper explains how the new SAS/ETS® COPULA procedure performs copula modeling and shows examples of using copula models for risk management problems.

INTRODUCTION

Measuring the performance and risk of holding financial assets is an important aspect of any good financial management plan. The value of each asset in a portfolio depends on a set of economic values called risk factors. These specific risk factors can impact the individual asset value, and they can impact the whole asset group. For example, the volatility of an individual stock is a risk factor that affects only assets that depend on that stock. On the other hand, the Federal Reserve deciding to increase interest rates impacts the whole banking sector although it is unlikely to impact other sectors such as manufacturing and agriculture. The severity of the impact might also depend on the individual positions of firms and their business plans. Clearly, a portfolio that holds a single asset has a higher risk exposure than one that holds a group of assets. Creating a well-balanced portfolio is an effective tool for managing risk.

A portfolio manager can use the fact that stock prices depend directly on individual volatilities and not directly on interest rates to generate what mathematically looks like a diversified portfolio. Unfortunately, the risk factors themselves are not independent. High stock volatility usually follows changes in interest rate policy.

The correlation of the risk factors in a portfolio can be measured and modeled in different ways. Many of these methods rely on simple distributional (normality) assumptions; if these assumptions hold, the obtained measures perform reasonably well. For example, if the underlying distributions are multivariate normal, you can use the correlation matrix to obtain a measure of joint asset movement. You can also use multivariate regression analysis to estimate the joint movement. However, in practice these simplifying assumptions often do not hold. Underlying marginal distributions might not be symmetric or normal, or they might have thicker tails. Any departure from normality of marginal distributions causes a problem for multivariate modeling that is based on the normality assumptions. It also causes the estimates of the dependent structure to become unreliable.

Copula functions were first introduced by Sklar (1959) and were used for years in survival analysis and actuarial sciences. Since the copula concept was first introduced in default modeling in risk management by Li (2000), it gained in popularity quickly and today is considered an essential tool in credit, market, and operational risk management modeling.

This paper explains how the new SAS/ETS COPULA procedure performs copula modeling on a hypothetical portfolio of four technology stocks.

COPULA CONCEPT

The word copula originates from Latin; it means to link, join, or connect. In a statistical sense, it refers to linking univariate marginal distributions to create a joint multivariate distribution.

The concept of copula is summarized by the Sklar (1959) theorem, which claims that any joint cumulative multivariate distribution function can be derived in terms of a copula and its continuous marginal distributions. Sklar defines the copula function as follows:

- Let F be a joint distribution function and $F_j, j = 1, \dots, m$ be the marginal distributions. Then there exists a copula $C : [0, 1]^m \rightarrow [0, 1]$ such that

$$F(x_1, \dots, x_m) = C(F_1(x_1), \dots, F_m(x_m))$$

for all x_1, \dots, x_m in $[-\infty, \infty]$. Moreover, if the margins are continuous, then C is unique; otherwise C is uniquely determined on $\text{Ran}F_1 \times \dots \times \text{Ran}F_m$, where $\text{Ran}F_j = F_j([-\infty, \infty])$ is the range of F_j .

- The converse is also true. That is, if C is a copula and F_1, \dots, F_m are univariate distribution functions, then the multivariate function defined in the preceding equation is a joint distribution function with marginal distributions $F_j, j = 1, \dots, m$.

The new COPULA procedure enables you to use the copula concept to fit and simulate multivariate distributions. The following copula types are supported:

- normal copula
- Student's t copula
- Clayton copula
- Gumbel copula
- Frank copula

In the following sections, all five supported copulas are discussed in more detail.

NORMAL COPULA

Let $u_j \sim U(0, 1)$ for $j = 1, \dots, m$, where $U(0, 1)$ represents the uniform distribution on the $[0, 1]$ interval. Let Σ be the correlation matrix with $m(m - 1)/2$ parameters satisfying the positive semidefiniteness constraint. The normal copula can be written as

$$C_{\Sigma}(u_1, u_2, \dots, u_m) = \Phi_{\Sigma}(\Phi^{-1}(u_1), \dots, \Phi^{-1}(u_m))$$

where Φ is the distribution function of a standard normal random variable and Φ_{Σ} is the m -variate standard normal distribution with mean vector 0 and covariance matrix Σ . That is, the distribution Φ_{Σ} is $N_m(0, \Sigma)$.

STUDENT'S t COPULA

Let $\Theta = \{(\nu, \Sigma) : \nu \in (1, \infty), \Sigma \in \mathbb{R}^{m \times m}\}$ and let t_{ν} be a univariate t distribution with ν degrees of freedom.

The Student's t copula can be written as

$$C_{\Theta}(u_1, u_2, \dots, u_m) = \mathbf{t}_{\nu, \Sigma}(t_{\nu}^{-1}(u_1), t_{\nu}^{-1}(u_2), \dots, t_{\nu}^{-1}(u_m))$$

where $\mathbf{t}_{\nu, \Sigma}$ is the multivariate Student's t distribution with a correlation matrix Σ with ν degrees of freedom.

ARCHIMEDEAN COPULAS

Let function $\phi : [0, 1] \rightarrow [0, \infty)$ be a strict Archimedean copula generator function and suppose its inverse ϕ^{-1} is completely monotonic on $[0, \infty)$. A strict generator is a decreasing function $\phi : [0, 1] \rightarrow [0, \infty)$ that satisfies $\phi(0) = \infty$ and $\phi(1) = 0$. A decreasing function $f(t) : [a, b] \rightarrow (-\infty, \infty)$ is completely monotonic if it satisfies

$$(-1)^k \frac{d^k}{dt^k} f(t) \geq 0, k \in \mathbb{N}, t \in (a, b)$$

An Archimedean copula is defined as follows:

$$C(u_1, u_2, \dots, u_m) = \phi^{-1}(\phi(u_1) + \dots + \phi(u_m))$$

The Archimedean copulas available in the COPULA procedure are the Clayton copula, the Frank copula, and the Gumbel copula.

Clayton Copula

Let the generator function be $\phi(u) = \theta^{-1}(u^{-\theta} - 1)$. A Clayton copula is defined as

$$C_{\theta}(u_1, u_2, \dots, u_m) = \left[\sum_{i=1}^m u_i^{-\theta} - m + 1 \right]^{-1/\theta}$$

with $\theta > 0$.

Frank Copula

Let the generator function be

$$\phi(u) = -\log \left[\frac{\exp(-\theta u) - 1}{\exp(-\theta) - 1} \right]$$

A Frank copula is defined as

$$C_{\theta}(u_1, u_2, \dots, u_m) = \frac{1}{\theta} \log \left\{ 1 + \frac{\prod_{i=1}^m [\exp(-\theta u_i) - 1]}{[\exp(-\theta) - 1]^{m-1}} \right\}$$

with $\theta \in (-\infty, \infty)$ for $m = 2$ and $\theta > 0$ for $m \geq 3$.

Gumbel Copula

Let the generator function be $\phi(u) = (-\log u)^{\theta}$. A Gumbel copula is defined as

$$C_{\theta}(u_1, u_2, \dots, u_m) = \exp \left\{ - \left[\sum_{i=1}^m (-\log u_i)^{\theta} \right]^{1/\theta} \right\}$$

with $\theta > 1$.

MODELING WITH THE COPULA PROCEDURE

This paper uses a hypothetical portfolio that consists of four technology stocks: Apple Inc. (AAPL), Dell Inc. (DELL), Hewlett-Packard Company (HPQ), and International Business Machines Corporation (IBM). The data were collected daily from January 2, 1990, to December 3, 2010, and represent the closing price for each given stock on the New York Stock Exchange (NYSE) or NASDAQ.


You use the following steps to model the risk exposure of these four stocks. For convenience, the complete, uninterrupted set of statements is contained in the Appendix.

1. Use a DATA step to difference and log the stock prices.
2. Use a macro with PROC COPULA to fit and simulate Clayton, Frank, Gumbel, normal, and Student's t copulas.
3. Use a macro with the RANK procedure to calculate empirical copulas.
4. Use a DATA step to merge the results for individual copulas.
5. Use the UNIVARIATE procedure to calculate percentiles of the empirical copulas.
6. Use the SGPLOT procedure to plot empirical copulas.

The following subsections describe each step in greater detail.

STEP 1: DATA PREPARATION


Since all four stocks are from the technology sector, they are exposed to the same external conditions. These conditions result in a trend that is the same for the whole sector; however, large performance differences exist among the individual stocks as shown in [Output 1](#).

Output 1 Original Data Series for Four Technology Stocks

The following statements difference the prices for all four stocks and perform a logarithmic transformation to calculate daily returns in percentages:

```
aapl_ret = log(aapl) - log(lag(aapl));  
dell_ret = log(dell) - log(lag(dell));  
ibm_ret = log(ibm) - log(lag(ibm));  
hpq_ret = log(hpq) - log(lag(hpq));  
run;
```

The differenced series are plotted in [Output 2](#).

Output 2 Differenced Data Series for Four Technology Stocks

The total number of days in the sample is 5,458. The number of days with losses outweighs the number of days with gains for all four stocks. [Table 1](#) shows the number of days with losses for all four stocks. The losses and gains ranged from -73% to 29%.


Table 1 Number of Days with Losses for the Portfolio

Stock	Number of Days with Losses
AAPL	2,494
DELL	2,493
IBM	2,540
HPQ	2,551

The Sklar theorem shows clearly that marginal distributions play an essential role in copula modeling and that the choice of a correct marginal distribution is important. The marginal distributions can be evaluated using a histogram in

which the bins with gains and losses are plotted against their proportions. [Output 3](#) shows a panel of histograms for all four stocks.

Output 3 Histogram of Differenced Data Series for Four Technology Stocks



You can see that the marginal distributions are symmetric and centered around zero. A normal or a t distribution would be a good candidate to describe the individual stock returns.

STEP 2: COPULA FITTING AND SIMULATION

The following statements fit a copula and generate a data set with 5,000 simulations:

```
proc copula data=tech;
  var aapl_ret dell_ret ibm_ret hpq_ret;
  fit t /marginals=empirical;
  simulate /ndraws = 5000
    seed = 12345678
    marginals=empirical
    plots=(data=original matrix)
    out = returns_t;
run;
```

In this example, the t copula is fitted; however, the FIT statement can be easily altered to fit the Clayton, Frank, Gumbel, or normal copulas. The SIMULATE statement simulates 5,000 observations. The MARGINALS=EMPIRICAL option in the FIT statement indicates that the uniform marginal distributions used in the copula calculation are created by using the empirical cumulative distribution functions (CDFs) based on data in the tech data set. The obtained random samples are output to the simulated_returns data set. The MARGINALS=EMPIRICAL option in the SIMULATE statement implies that the marginal distributions output to the simulated_returns data set were obtained by using the inverse of an empirical CDF computed from the DATA= option in the PROC COPULA statement.

The parameter estimates of Θ for the Archimedean family of distributions and the degrees of freedom for the t copula are shown in Table 2. All four estimates of Θ and DF are significant at the 1% level. The estimates of the correlation matrix for the normal and t copulas are shown in Table 3 and Table 4.

Table 2 Copula Parameter Estimates

Distribution	Parameter	Estimate	Standard Error	t Value	p Value
Clayton	Θ	0.52545	0.01078	48.75	0.0001
Frank	Θ	2.62555	0.04683	56.06	0.0001
Gumbel	Θ	1.34071	0.00768	174.79	0.0001
t	DF	5.43512	0.22704	23.94	0.0001


Table 3 Correlation Matrix Estimates for the Normal Copula

Variable	AAPL	DELL	IBM	HPQ
AAPL	1.000000000	0.415582950	0.378399439	0.398248951
DELL	0.415582950	1.000000000	0.408624646	0.434584401
IBM	0.378399439	0.408624646	1.000000000	0.475411735
HPQ	0.398248951	0.434584401	0.475411735	1.000000000

Table 4 Correlation Matrix Estimates for the t Copula

Variable	AAPL	DELL	IBM	HPQ
AAPL	1.000000000	0.430353924	0.402895360	0.414602148
DELL	0.430353924	1.000000000	0.439244257	0.462049749
IBM	0.402895360	0.439244257	1.000000000	0.502548281
HPQ	0.414602148	0.462049749	0.502548281	1.000000000

The PLOTS=(DATA=ORIGINAL MATRIX) option plots the simulated marginal distributions of all four stocks in a matrix form for t copula. The plots are shown in Output 4.

Output 4 Simulated Data**STEP 3: EMPIRICAL COPULA AND COPULA COMPARISON**

Which copula is the best? It is not sufficient to examine the parameters in Table 2, Table 3, and Table 4 and the simulated data in simulated_returns. One way to compare the results is to estimate empirical copulas for the original and the simulated data sets.

Deheuvels (1979, 1981) first introduced the concept of the empirical copula and showed that the empirical copula converges uniformly to the underlying parametric copula. If you define $X_t = \{X_{1t}, X_{2t}, \dots, X_{nt}\}$ as an independent and identically distributed (i.i.d.) sequence with continuous CDF F and continuous marginal distributions F_j , the empirical copula is defined as

$$\hat{C}\left(\frac{t_1}{T}, \frac{t_2}{T}, \dots, \frac{t_n}{T}\right) = \frac{1}{T} \sum_{t=1}^T \prod_{j=1}^n \mathbf{1}(r_j^t \geq t_j)$$

where $\mathbf{1}$ is an indicator function that takes the value 1 when the condition is satisfied, and r_j^t is a rank statistic of variable X_t . In a less formal way, the empirical copula can be defined as a rank statistic for each observation divided by the total number of observations. To compare the five multivariate distributions, use the RANK procedure to calculate the empirical copula for the original data and all five simulations, find the smallest rank for all four stocks and sort the data set in the ascending order. The following statements calculate the empirical copula using the simulated data for the Student's t distribution:

```
data rank;
set returns_t;
proc rank data=rank out=order descending fraction;
var aapl_ret dell_ret ibm_ret hpq_ret;
ranks aapl_rank dell_rank ibm_rank hpq_rank;
run;

data emp_copula_t;
set order;
keep rank_t;
rank_t = min(aapl_rank, dell_rank, ibm_rank, hpq_rank);
run;

proc sort data=emp_copula_t;
by rank_t;
run;
```

STEP 4: MERGING THE RESULTS FOR INDIVIDUAL COPULAS AND THE PORTFOLIO

The following statements merge the empirical copulas for the portfolio and all five models.

```
data merged;
merge emp_copula_tech emp_copula_clayton emp_copula_frank
emp_copula_gumbel emp_copula_normal emp_copula_t;
run;
```

STEP 5: PERCENTILE CALCULATION USING THE UNIVARIATE PROCEDURE

The following statements use the UNIVARIATE procedure to calculate percentiles for each copula and the original data:

```
proc univariate data=merged;
var rank_tech rank_clayton rank_frank rank_gumbel rank_normal rank_t;
output out=Pctls pctlpts=5 to 100 by 5 pctlpre=pctlpre=t_tech t_clayton
t_frank t_gumbel t_normal t_t
pctlname=P5 P10 P15 P20 P25 P30 P35 P40 P45 P50 P55 P60 P65
P70 P75 P80 P85 P90 P95 P100;
run;
```

The results of the UNIVARIATE procedure for the original data and all five simulated copulas are shown in [Table 5](#).

Table 5 Portfolio and Simulated Data Percentiles

Percentile	Portfolio	Simulated Data				
		Clayton	Gumbel	Frank	Normal	t
0.05	0.01521	0.01319	0.01319	0.01814	0.01521	0.01741
0.1	0.03316	0.02675	0.02767	0.03664	0.03096	0.03554
0.15	0.05167	0.04049	0.04397	0.05460	0.04892	0.05515
0.2	0.07164	0.05606	0.06174	0.07604	0.06816	0.07512
0.25	0.09271	0.07420	0.08153	0.09656	0.08776	0.09674
0.3	0.11506	0.09271	0.10132	0.11927	0.10975	0.12074
0.35	0.14181	0.11158	0.12569	0.14181	0.13283	0.14786
0.4	0.16801	0.13338	0.15097	0.16526	0.15940	0.17424
0.45	0.19384	0.15812	0.18139	0.19091	0.18560	0.20117
0.5	0.22472	0.18303	0.21134	0.21867	0.21400	0.23150
0.55	0.25797	0.21088	0.24551	0.24588	0.24643	0.26310
0.6	0.29498	0.24203	0.28234	0.27629	0.28417	0.29718
0.65	0.33510	0.27684	0.31990	0.31257	0.32393	0.33749
0.7	0.38421	0.31605	0.36094	0.35288	0.36424	0.38329
0.75	0.43734	0.36387	0.40876	0.39337	0.41059	0.43056
0.8	0.53261	0.41957	0.46189	0.43936	0.46574	0.48241
0.85	0.56559	0.48882	0.52327	0.49066	0.53188	0.55460
0.9	0.65464	0.58428	0.59307	0.56376	0.60993	0.63137
0.95	0.76603	0.71473	0.68798	0.66490	0.72078	0.74386
1	0.99634	1.00000	0.94650	0.93661	0.99707	0.98516
$(K_n^P(t) - K_n^S(t))^2$	-	0.05111	0.02191	0.0.04042	0.01156	0.00414

$(K_n^P(t) - K_n^S(t))^2$ represents the squared differences between the empirical copulas based on the original data and simulations.



STEP 6: PLOTTING THE RESULTS

The following statements use the SGPLOT procedure to create two sets of plots:

```
proc sgplot data=sgf.merged;
  series x=rank_tech y=rank_tech;
  series x=rank_tech y=rank_clayton;
  series x=rank_tech y=rank_frank;
  series x=rank_tech y=rank_gumbel;
  xaxis label="Portfolio";
  yaxis label="Simulation";
run;

proc sgplot data=sgf.merged;
  series x=rank_tech y=rank_tech;
  series x=rank_tech y=rank_normal;
  series x=rank_tech y=rank_t;
  xaxis label="Portfolio";
  yaxis label="Simulation";
run;
```

Output 5 shows the results for the simulation of normal and t copulas, and Output 6 contains the results for the Clayton, Frank, and Gumbel copulas. The empirical copula based on the original data is plotted along the X axis, and the empirical copulas based on the simulation are plotted along the Y axis. The blue line is the 45 degree reference line; it shows a perfect fit.

Output 5 Empirical Copula Plots for Normal and t Distributions**Output 6** Empirical Copula Plots for Clayton, Frank, and Gumbel Distributions

Output 2 shows that the data are heavily concentrated around zero returns. Table 6 shows the number of days with zero returns for all four stocks.

Table 6 Number of Days with Zero Returns for the Portfolio

Stock	Number of Days with Zero Returns
AAPL	359
DELL	331
IBM	287
HPQ	298

The heavy concentration around zero returns is also reflected in the simulation plots in [Output 5](#) and [Output 6](#), where it is demonstrated by vertical and horizontal segments in a zig-zag pattern. The rank for data points in the vertical or horizontal segments is not necessarily the same, but the differences cannot be seen due to graph scaling.

SUMMARY

The t empirical copula with the $DF = 5.43$ shown in [Output 5](#) is the overall winner. It matches the original data well, and the green line for the t copula is closest to the reference line. The measure of overall fit for the t copula, $(K_n^P(t) - K_n^S(t))^2$, is 0.00414. The normal copula has the second-best fit. It is represented by the red line, and its measure of overall fit, $(K_n^P(t) - K_n^S(t))^2$, is 0.01156.

The Archimedean family of copulas is shown in [Output 6](#). The Frank copula is in the overall third place, and it fits the best in the Archimedean family. Frank and Clayton copulas have the largest departures from their empirical copulas based on the real data. The Clayton copula fits well in the upper tail, but overall it provides the poorest fit out of all tested copulas.

CONCLUSION

In recent years, the copula modeling has become a popular tool in the area of risk management. It allows a great flexibility in estimating multivariate distributions and eliminates the need for simplifying assumptions. This paper demonstrates the use of a new SAS/ETS procedure, the COPULA procedure. PROC COPULA is used to fit and simulate data from several multivariate distributions. The results are examined and compared using empirical copula fitted to the original and simulated series. The overall winner is a copula fitted with a t distribution. The empirical copula for the simulated data with t distribution is closest to the original series, and it has the smallest measure of overall fit.

REFERENCES

- Deheuvels, P. (1979), "La Fonction de Dependance Empirique et Ses Proprietes-Un Test Non Parametrique d'Independance," *Academie Royale de Belgique, Bulletin de la Classe des Sciences, 5e Serie*, 65, 274–292.
- Deheuvels, P. (1981), "A Non Parametric Test for Independence," *Publications de l'Institut de Statistique de l'Universite de Paris*, 26, 29–50.
- Erdman, D. and Arthur Sinko (2008), "Using Copulas to Model Dependency Structures in Econometrics," *Proceedings of the SAS Global Forum 2008 Conference*, Paper 321-2008.
- Li, David X. (2000), "On Default Correlation: A Copula Function Approach," *Journal of Fixed Income*, 9: 43–54.
- Sklar, A. (1959), "Fonctions de Répartition à n Dimensions et Leurs Marges," *Publications de l'Institut de Statistique de l'Université de Paris*, 8, 229–231.

APPENDIX

The following statements fit and simulate copulas, calculate empirical copulas, and plot the empirical copula results:

```

data tech;
set sgf.bloomberg;
format date date9.;
keep date aapl dell ibm hpq aapl_ret dell_ret ibm_ret hpq_ret;

/* Difference and log the stock prices ---*/
aapl_ret = log(aapl)-log(lag(aapl));
dell_ret = log(dell)-log(lag(dell));
ibm_ret = log(ibm)-log(lag(ibm));
hpq_ret = log(hpq)-log(lag(hpq));
run;

data returns_tech;
set tech;
keep aapl_ret dell_ret ibm_ret hpq_ret;
run;

/* Macro to fit and simulate */
%macro copula_fit_simulate(name, seed);
proc copula data=tech;
var aapl_ret dell_ret ibm_ret hpq_ret;
fit &name /marginals=empirical;
simulate /ndraws = 5458
seed = &seed
out = returns_&name;
;
run;
%mend;

/*Macro to calculate empirical copula */
%macro copula_rank(name);
data rank;
set returns_&name;

proc rank data=rank out=order descending fraction;
var aapl_ret dell_ret ibm_ret hpq_ret;
ranks aapl_rank dell_rank ibm_rank hpq_rank;
run;

data emp_copula_&name;
set order;
keep rank_&name;
rank_&name = min(aapl_rank, dell_rank, ibm_rank, hpq_rank);
run;

proc sort data=emp_copula_&name;
by rank_&name;
run;

%mend;

%copula_fit_simulate(name=clayton, seed=1234);
%copula_fit_simulate(name=frank, seed=12345);
%copula_fit_simulate(name=gumbel, seed=123456);
%copula_fit_simulate(name=normal, seed=1234567);
%copula_fit_simulate(name=t, seed=12345678);

%copula_rank(name=tech);
%copula_rank(name=clayton);

```

```

%copula_rank(name=frank);
%copula_rank(name=gumbel);
%copula_rank(name=normal);
%copula_rank(name=t);

data merged;
merge emp_copula_tech emp_copula_clayton emp_copula_frank
      emp_copula_gumbel emp_copula_normal emp_copula_t;
run;

/* Calculation of percentiles */
proc univariate data=merged;
  var rank_tech rank_clayton rank_frank rank_gumbel rank_normal rank_t;
  output out=Pctls pctlpts=5 to 100 by 5 pctlpre=pctlpre=t_tech t_clayton
        t_frank t_gumbel t_normal t_t
        pctlname=P5 P10 P15 P20 P25 P30 P35 P40 P45 P50 P55 P60 P65
        P70 P75 P80 P85 P90 P95 P100;
run;

/* Plot empirical copulas against the tech portfolio */
proc sgplot data=merged;
series x=rank_tech y=rank_tech;
series x=rank_tech y=rank_clayton;
series x=rank_tech y=rank_frank;
series x=rank_tech y=rank_gumbel;
xaxis label="Portfolio";
yaxis label="Simulation";
run;

proc sgplot data=merged;
series x=rank_tech y=rank_tech;
series x=rank_tech y=rank_normal;
series x=rank_tech y=rank_t;
xaxis label="Portfolio";
yaxis label="Simulation";
run;

```

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

Jan Chvosta
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.